BCs Thesis

Cracow, 2015

CONSIDERING GRID CONSTRAINS IN ENERGY MODELS

Javier Ruiz-Herrera Poyal

Supervisor:

Dr inż. Artur Wyrwa

Faculty of Energy and Fuels

1. Abstract 3
2. AC Power Flow 4
3. DC Linearized Power Flow 5
    3.1. Assumptions 5
    3.2. Equations 6
    3.3. Analysis of error 7
4. Model of the Polish grid 8
    4.1. Criteria for the selection of nodes 9
    4.2. Criteria for the selection of power lines 11
5. Parameters of the model 17
    5.1. Lines reactance 17
    5.2. Demand 18
    5.3. Generation 20
6. Application to energy-economic studies 23
7. Simulation 24
    7.1. Randomization of the input 24
    7.2. Sensitivity analysis 25
8. Conclusions 30
    8.1. Further improvement 30

References 31
List of figures 32
GAMS code 33

1. Abstract

Energy-economic models are an important tool employed in the assessment of investments for new power capacity. These are purely operational models, taking into consideration market clearing conditions and power plant constrains (such as generation limits, fuel availability, emission costs…) in order to minimize the total cost of the system.

However, energy-economic models fail to consider the physical limitations of the electrical grid. The disregard for this fact may result in the phenomena known as congestion, a shortage of transmission capacity in the grid, which ultimately implies an economic loss. This is the reason why it is necessary to include the use of a physical model in this kind of studies.

Poland serves as a great example of risk of congestion, due to its relatively old network, its unbalance of generation and demand between different regions and the numerous studies being conducted in order to assess the installation of new capacity, especially renewable sources such as wind.

A physical model employs the features of the power lines, the generation, the demand, and the allocation of installed capacity as inputs in order to study the power flows and, in the end, to obtain the optimal allocation of future power plants as an output.

An existing network model is the AC power flow. However, due to its extreme complexity, it is not employed in energy-economic models. Fortunately, through a series of simplifications it is possible to obtain a DC power flow model, a linearization of the AC power flow, much simpler and with an acceptable margin of error.

In this work we will employ the modelling software GAMS to model the set of equations necessary to build a DC linear power flow system. Then we aim to represent with a satisfactory level of accuracy the Polish grid and to use this model with past energy-economic studies in order to observe how the optimal mix of fuels to be installed will vary when taking into consideration the physical constrains of the grid.

2. AC Power Flow

The electrical grid, as a physical system, is subject to the laws of electricity. It is a necessity in order to simulate and study the behaviour of this system to understand the equations that rule the network.

The sum of all the complex power at a determined node must be equal to zero. Understanding that complex power (S_i) is composed as the sum of real power (P_i) and reactive power (Q_i) in the following way $S_i = P_i + j \cdot Q_i$, the conservation of power at node i connected to j neighbouring nodes would be expressed as follows:

$$P_i = \sum_j |V_i||V_j|(G_{ij} \cos \theta_{ij} + B_{ij} \sin \theta_{ij})$$

$$Q_i = \sum_j |V_i||V_j|(G_{ij} \sin \theta_{ij} - B_{ij} \cos \theta_{ij})$$

which has the following unknown variables:

$|V_i|$: Voltage magnitude of node i

$\theta_{ij} = \delta_i - \delta_j$: Difference between the phase angles of neighbouring nodes

P_i: Resulting real power at node i

Q_i: Resulting reactive power at node i

And requires the following physical features of the grid as inputs:

G_{ij}: Conductance of the line

B_{ij}: Susceptance of the line

This composes a non-linear system which proves to be of great complexity and not an efficient tool in the study of energy models, due to the high computational power involved in solving the iterative mathematical methods.

3. DC Linearized Power Flow

3.1. Assumptions

Fortunately, it is possible to obtain a linear simplification of the equations that allows us to solve the system in an effective way, in exchange for a certain error that we will later address. This is known as the "DC Linear Power Flow Equations" and it is achieved through the following assumptions:

1. Line resistances are negligible compared to line reactances. As a consequence, grid losses are neglected and line parameters are simplified.

$$R_l \ll X_l \text{ for all lines}$$

$$P_i = \sum_j |V_i||V_j|B_{ij} \sin \theta_{ij}$$

$$Q_i = \sum_j |V_i||V_j|(-B_{ij} \cos \theta_{ij})$$

2. Voltage phase angles of neighbouring nodes are similar. This means that the sine of the difference can be approximated by the difference of the angles themselves and that the cosine of the difference will be close to 1.

$$P_i = \sum_j |V_i||V_j|B_{ij}\theta_{ij}$$

$$Q_i = \sum_j |V_i||V_j|(-B_{ij})$$

3. The voltage is considered flat, i.e. the voltage amplitude in *per-unit* is the same across all the nodes and equal to 1.

$$|V_i| = 1\, p.u. \text{ for every node}$$

And the equations result in:

$$P_i = \sum_j B_{ij}\theta_{ij}$$

Where, with further analysis it can be proven that:

$$P_{ij} \gg Q_{ij}$$

And thus we can consider only active power flows in our model.

3.2.Equations

After applying the simplifying assumptions to the AC Equations, we obtain the DC Linearized Power Flow Equations for a transmission line from node *i* to node *j*:

$$P_{ij} = B_{ij}(\delta_i - \delta_j) = (\delta_i - \delta_j)/X_{ij}$$

And in every node we can conduct a power balance:

$$G_i - Q_i = \sum_l P_l$$

X_{ij}: Reactance of the line

G_i: Generation of power injected in node *i*

Q_i: Consumption of power in node *i*

As a result, our system is composed by $n+L$ equations and $n+L$ unknown variables, which usually means it's a determined system but the nodal balances are actually linearly dependent so the useful number of equations will be $n+L-1$. However, given that the set of unknown variables for the phase angles is only expressed as differences, it is necessary to establish a reference point, for which we add an extra equation for the reference node with $\delta_{ref} = 0$. And thus, our system will be defined with $n+L$ equations and $n+L$ unknown variables.

3.3.Analysis of error

Of course, as in any other simplification, there is a sacrifice in accuracy as a result of every assumption made. While the linearized model is an inestimable tool in energy studies, it is important to be aware of its limitations.

Line reactances are negligible compared to line reactances. In real scenarios, the ration x/r is in the range between 2 and 10. The highest this ratio is, the more valid this assumption is. For ratios higher than 2 the average error will always be smaller than 5% and for ones above 5, it will be below 2% in average.

Voltage phase angles of neighbouring nodes are similar. In most cases the difference between neighbouring nodes (i.e., ones connected by a power line) will be less than 15º degrees, and it is very rare to see a difference above 30º. This assumption is more accurate if the grid is weakly loaded and less reliable during load peaks. But even in this case, and only in the lines affected by the peak, the error caused by this assumption is less than 1%.

The voltage is considered flat. The per-unit value of voltage in most operating conditions is between 0.95 and 1.05. Most standard deviations will be of the order of 0.01 p.u., which produces an average error of approximately 5%. However in real scenarios it is usual to exceed this amount, and thus making this assumption the most important source of error of the DC Linearized model.

As conclusion, while the DC model can have a high error for the study of separate single lines, for the whole grid on average the error will be of around 5% when compared to the AC model. However, the AC model will also have a non-negligible error with respect to the real grid due to simplifications of the configuration and input data.

4. Model of the Polish grid

Now that we understand the equations that are going to run our model, it is time to consider how to obtain the necessary parameters that are involved in the system. This is going to be highly dependent on the disposition we select to represent the real grid, i.e., the number of nodes, their location, the lines connecting them, and the generation and demand assigned to them.

This will constitute a simplified grid model with no exports or imports and, as a result, the generation and the load in the Polish territory will be equal.



Figure 1. Transmission map of the Polish grid

4.1. Criteria for the selection of nodes

There are several things that are worth to account for while selecting the location of a node. The more strictly these guidelines are followed, the more reliable our system will be.

Firstly, it should be close to as many electrical substations (i.e., high voltage transformers) as possible as this will allow us to connect the node to a greater number of power lines.

Secondly, the power plants that will constitute the generation should have a node on its location, and if this is not practical, as close as possible. This will make the premise of power being injected into the node more accurate.

And lastly but equally important, the nodes should take into account the way that is going to be used to estimate the demand. If the population around the nodes is going to be used, they should be located on zones of high population density whenever is possible. In our case, the demand will be estimated through the peak demand of the different regions of Poland. For this reason, the model aimed to have a node in almost every region, unless the high voltage power lines of two regions can be aggregated into one.

With all of this in mind, we allocated 12 different nodes in the following way:



Figure 2. Allocation of nodes on the map

Region	n1	n2	n3	n4	n5	n6	n7	n8	n9	n10	n11	n12
Dolnośląskie							•					
Kujawsko-Pomorskie			•									
Łódzkie								•				
Lubelskie											•	
Lubuskie							•					
Małopolskie									•			
Mazowieckie					•							
Opolskie									•			
Podkarpackie										•		
Podlaskie						•						
Pomorskie	•											
Śląskie									•			
Świętokrzyskie										•		
Warmińsko-Mazurskie												•
Wielkopolskie				•								
Zachodniopomorskie		•										

Figure 3. Table linking nodes with Polish regions

4.2.Criteria for the selection of power lines

Once we have the nodes placed it is time to establish the connections between them with transmission lines. We will take into account only the high voltage grid, i.e. lines of 220 kV and 400 kV.

First it is important to define two concepts that will be frequently used in our model:

Transmission line: the connection between two nodes. It can be constituted of one or several circuits in parallel with different voltage.

Circuit: A single 3-phase circuit connecting two nodes.

In order to make the model more reliable it is important to include as many lines as possible. There are several reasons why an existing line may not be included in the model:

- The line connects two stations that are considered to be in the same node. This is the main cause of exclusion, especially in zones with a lot of generation and demand very concentrated, like the surroundings of Katowice (node 9).

- There is not any node at the beginning or end of the line. In a perfect model every single electrical substation would constitute a node, being highly more complex than what our needs demand. After 12 nodes, adding another node would only produce an increase of around 100 km in modelled lines, which would mean roughly a 1% increase of the total length included.

- The line is connecting Poland with the neighbouring countries. As we previously said, exports and imports are not in the scope of the model.

With all the considerations taken into account, we obtain the following scheme of our model:



Figure 4. Circuits displayed on the map



Figure 5. Scheme of transmission lines

Line	Start	End	Circuit	Voltage [kV]	Distance [km]	Stations
l1	n1	n2	c1	400	352.60	ZRC-SLK-DUN-MON-KRA
l2	n1	n3	c1	400	114.00	GBL-GRU
l2	n1	n3	c2	220	185.58	GDA-JAS
l3	n1	n4	c1	220	376.47	GDA-ZYD-PKW-PLE

l4	n1	n12	c1	400	153.77	GBL-OLM
l5	n2	n4	c1	400	241.26	KRA-PLE
l6	n2	n7	c1	220	323.44	KRA-GOR-LSN-MIK
l7	n3	n4	c1	220	100.74	JAS-PAT
l7	n3	n4	c2	220	100.74	JAS-PAT
l7	n3	n4	c3	220	137.86	TEL-WLA-PAT
l8	n3	n5	c1	400	304.88	GRU-PLO-MIL
l9	n3	n12	c1	220	270.42	OLS-WLA-TEL
l10	n4	n5	c1	220	265.11	PAT-PDE-MOR
l10	n4	n5	c2	220	182.93	KON-SOC-OLT-MOR
l11	n4	n7	c1	400	405.62	PLE-KRM-OSR-PAS-CRN-MIK
l11	n4	n7	c2	220	320.79	PLE-LES-POL-MIK
l11	n4	n7	c3	220	320.79	PLE-LES-POL-MIK
l12	n4	n8	c1	400	129.91	OSR-ROG
l12	n4	n8	c2	220	217.40	KON-ADA-ZGI-JAN-ROG
l12	n4	n8	c3	220	190.89	KON-ADA-PAB-JAN-ROG
l13	n4	n9	c1	400	400.33	OSR-TRE-DBN-WIE
l14	n5	n6	c1	400	174.98	MIL-NAR
l15	n5	n8	c1	400	185.58	MSK-ROG
l15	n5	n8	c2	400	220.05	PLO-ROG
l15	n5	n8	c3	220	235.96	MOR-JAN-PAB-ROG
l16	n5	n10	c1	400	288.98	MIL-KOZ-OSC-PEL-RZE
l16	n5	n10	c2	220	448.05	MOR-KOZ-ROZ-KIE-PEL-CHM-BGC

l17	n5	n11	c1	400	106.05	KOZ-LSY
l17	n5	n11	c2	220	408.28	MOR-KOZ-PUL-ABR-MKR-CHS
l18	n5	n12	c1	220	267.77	OLS-OST-MIL
l19	n7	n9	c1	400	299.59	SWI-WRC-PAS-DBN-WIE
l19	n7	n9	c2	220	474.55	MIK-SWI-ZBK-GRO-KED-WIE-KOP
l20	n8	n9	c1	400	211.50	ROG-TCN-LAG-ROK-WIE
l20	n8	n9	c2	400	204.14	ROG-JOA-WIE
l20	n8	n9	c3	220	156.42	ROG-JOA-LAG
l20	n8	n9	c4	220	182.93	ROG-JOA-LOS-KHK-BYC
l21	n9	n10	c1	400	299.58	TCN-RZE
l21	n9	n10	c2	400	278.37	TCN-TAW-RZE
l21	n9	n10	c3	220	243.91	BYC-SIE-KLA-PEL-CHM-STW-ABR
l21	n9	n10	c4	220	182.93	BYC-SKA-KLA
l22	n10	n11	c1	220	167.03	PEL-CHM-STW-ABR

Figure 6. Characteristics of transmission lines

5. Parameters of the model

5.1. Lines reactance

The reactance of a circuit is solely dependent on its distance and voltage. To determine the reactance per unit of distance we will employ an interpolation of the following table:

Voltage [kV]	230	345	500	765
Resistance [Ω/m]	0.050	0.037	0.028	0.012
Reactance [Ω/m]	0.407	0.306	0.271	0.274
Admittance [μS/km]	2.764	3.765	4.333	4.148

Figure 7. Typical values of transmission lines parameters

Once we have determined the reactances of all the circuits we express them in a *per-unit* system, employing the biggest reactance of the grid:

$$x_c = \frac{X_c}{X_{c,max}}$$

And then we aggregate the circuit reactances into total line reactances:

$$x_l = \frac{1}{\sum_c \frac{1}{x_c}}$$

Line	Start	End	Reactance [p.u]
l1	n1	n2	0.5726
l2	n1	n3	0.1256
l3	n1	n4	0.7933

l4	n1	n12	0.2497
l5	n2	n4	0.3918
l6	n2	n7	0.6816
l7	n3	n4	0.0777
l8	n3	n5	0.4951
l9	n3	n12	0.5698
l10	n4	n5	0.2281
l11	n4	n7	0.2234
l12	n4	n8	0.1063
l13	n4	n9	0.6501
l14	n5	n6	0.2842
l15	n5	n8	0.1230
l16	n5	n10	0.3135
l17	n5	n11	0.1435
l18	n5	n12	0.5643
l19	n7	n9	0.3273
l20	n8	n9	0.0865
l21	n9	n10	0.1135
l22	n10	n11	0.3520

Figure 8. Reactances of transmission lines

5.2.Demand

For the demand, we used the peak load during 2011 (23,801 MW), as it is the most likely scenario to cause congestions of the grid, weighted by the total consumption of every region during the year.

Total Consumption [GWh]	n1	n2	n3	n4	n5	n6	n7	n8	n9	n10	n11	n12
Dolnośląskie							13232					
Kujawsko-Pomorskie			8317									
Łódzkie								11944				
Lubelskie											5694	
Lubuskie							3366					
Małopolskie									11315			
Mazowieckie					20601							
Opolskie									4942			
Podkarpackie										4943		
Podlaskie						2793						
Pomorskie	7969											
Śląskie									25454			
Świętokrzyskie										5091		
Warmińsko-Mazurskie												3490
Wielkopolskie				12879								
Zachodniopomorskie		5639										
Total	7969	5639	8317	12879	20601	2793	16598	11944	41711	10034	5694	3490
Peak load [MW]	1284	908.9	1341	2076	3320	450.2	2675	1925	6723	1617	917.7	562.4

Figure 8. Power consumption in nodes

For their inclusion in the code, these values will be expressed in *per-unit* with the power base given by the nominal power of the transformers, 730 VA.

5.3. Generation

As to characterize the generation we will use the installed capacity per region in 2011, only counting with the non-renewable power plants.

Installed Capacity [MW]	n1	n2	n3	n4	n5	n6	n7	n8	n9	n10	n11	n12
Dolnośląskie							2467					
Kujawsko-Pomorskie			731.7									
Łódzkie								5860				
Lubelskie											406.9	
Lubuskie							448.4					
Małopolskie									2080			
Mazowieckie					5103							
Opolskie									1832			
Podkarpackie										849		
Podlaskie						173.1						
Pomorskie	1238											
Śląskie									7420			
Świętokrzyskie										1597		
Warmińsko-Mazurskie												73.2
Wielkopolskie				2800								
Zachodniopomorskie		2226										
Total	1238	2226	731.7	2800	5103	173.1	2915	5860	11332	2446	406.9	73.2

Figure 9. Power generation in nodes

However, given that our model does not include imports or exports, market clearance must be fulfilled, i.e. production must be equal to demand. We will balance the generation in every node in the following way to meet this requirement:

$$G_i = G'_i \cdot \frac{\sum_i Q_i}{\sum_i G'_i}$$

Thus, this model does not take into account the priority of different power plants in order to inject their power into the grid.

For their inclusion in the code, these values will be again expressed in *per-unit*.

6. Application to energy-economic studies

Traditional studies for new power capacity employ solely an economic model to calculate the costs. However, this approach is lacking some physical considerations because it disregards the grid constrains; which leads to the risk of congestion, a shortage of transmission capacity.

Poland can be a great example of congestion due to the unbalance in installed power and demand between regions and to the many studies being conducted right now to increase capacity, especially considering renewable sources for the near future.

While the economic model takes into account inputs such as fuel costs, environmental constrains such as taxes on pollution, and the behaviour of the power plants; they fail to consider the grid physical features, like the maximum capacity, its behaviour, its cost or the geographic location of the resources. Through both mathematical models we can ensure the best output minimizing the costs.

To fulfil this purpose we develop a tool that will be able to assess, for a certain capacity that is going to be installed (composed of a certain mix of fuels), which allocation and distribution of its value over the Polish territory conforms a feasible system given the current state of the electrical grid.

A good approach for future expanding of this tool could include calculating the cost of every feasible scenario. This would take into account the different costs of generation and transport of energy depending on its location and as a result we would obtain the optimal scenario minimizing the costs.

7. Simulation

7.1. Randomization of the input

In order to assess the feasibility of many different scenarios, we developed a pseudo-randomizing algorithm. To explain said algorithm we will produce an example scenario for a new capacity of 1200 MW.

```
Cap = 1200;
```

We create a random vector which values can be 1, 2 or 3. This way we will create only 3 possible values equally distant to be installed in each node, meaning that the capacity installed in a node will not be unrealistically small or big:

```
ran(n) = uniformint(1,3);
```

We create a random binary vector. This allows us to set a random number of zeros that result in a 50% of average (If we included the 0 in the `ran(n)` vector we would only obtain a 25% of zeroes in average):

```
bin(n) = uniformint(0,1);
produ(n) = bin(n)*ran(n);
```

Now we just normalize `produ(n)` so the sum of its components results in `Cap`:

```
New_Gen(n) = Cap*produ(n)/Sum_produ;
```

A visual representation of the algorithm:

$$\begin{pmatrix} 1 \\ 3 \\ 2 \\ 1 \\ 1 \end{pmatrix} \cdot \begin{pmatrix} 0 \\ 1 \\ 1 \\ 0 \\ 1 \end{pmatrix} = \begin{pmatrix} 0 \\ 3 \\ 2 \\ 0 \\ 1 \end{pmatrix} \longrightarrow \begin{pmatrix} 0 \\ \frac{3 \cdot 1200}{6} \\ \frac{2 \cdot 1200}{6} \\ 0 \\ \frac{1 \cdot 1200}{6} \end{pmatrix} = \begin{pmatrix} 0 \\ 600 \\ 400 \\ 0 \\ 200 \end{pmatrix}$$

7.2. Sensitivity analysis

The most important parameters in the model are:

- `cap`: New capacity that is going to be installed in the system. It will condition the amount of new generation in every node, which may contribute to the unbalance between nodes and increase the power flow through the lines.

- `tcap`: Maximum capacity that line or circuit can transmit before reaching the thermal limit of the wire. It is the parameter that constrains the feasibility of the model. If a scenario can is unable to satisfy the demand without exceeding this value in a power line, the system is not feasible.

Thus, it is of great importance to assess how sensitive our model is to the variance of these two parameters. In order to conduct this assessment, we will run a simulation for 10 values of `cap`, 10 values of `tcap`, and 100 different random distributions of the capacity for each pair of values. We will then calculate the percentage of feasible scenarios in every situation to observe the influence of each parameter.

% Feasible Scenarios		New Capacity to be installed [MW]									
		1000	2000	3000	4000	5000	6000	7000	8000	9000	10000
Max. Transmission Capacity of a circuit of 400 kV [MW]	1400	100	100	100	99	97	91	93	87	82	70
	1300	100	100	100	99	99	93	90	82	77	51
	1200	100	100	100	98	96	90	87	75	64	48
	1100	100	100	99	100	94	87	80	60	65	50
	1000	100	99	99	97	88	76	66	51	48	35
	900	100	100	98	94	87	73	58	51	32	26
	800	100	99	98	90	75	56	45	36	33	23
	700	100	99	91	77	50	44	36	23	18	11
	600	78	74	50	43	41	20	20	17	5	7
	500	9	18	22	20	17	13	11	2	2	1

Figure 10. Percentage of feasible scenarios as a function of cap and tcap



Figure 11. Feasible scenarios as a function of cap for tcap=1400



Figure 12. Feasible scenarios as a function of cap for tcap=1000



Figure 13. Feasible scenarios as a function of cap for tcap=500



Figure 14. Feasible scenarios as a function of tcap for cap=1000



Figure 15. Feasible scenarios as a function of tcap for cap=5000



Figure 16. Feasible scenarios as a function of tcap for cap=10000

8. Conclusions

As it can be observed from the results, the transmission capacity tcap is the main factor to affect the output. This is especially true for the smaller values of cap, where any value for tcap above 800 MW will provide more than 90% feasible scenarios.

It can be concluded that the necessity for a study on the grid increases as the capacity to be installed increases and as the maximum power that the lines can transmit decreases. If an investment in capacity is located in the range of cap and tcap where it is convenient to assess its feasibility, a tool similar to this model conforms an easy and accessible way to acquire assurance.

8.1. Further improvement

For future development of this tool, the following features could be added:

- Implementation of a cost associated to every scenario. Evaluating the different costs of the distribution of new capacity would allow minimizing the cost as a goal function, obtaining the most efficient output.

- Establishing a priority system for the power production. Given that different power plants have different priorities at the time of injecting power in to the grid, this feature would make the system more reliable.

- Including transmission through the border. Including imports and exports would represent in a more realistic way the behaviour of the grid.

- Higher number of scenarios. Due to computational limitations we ran 100 random scenarios per value of `cap` and `tcap`, which is not large enough to obtain an accurate percentage of feasible scenarios due to the extremely high amount of different possible combinations for the allocation.

References

A. Stiel: “Modelling Liberalised Power Markets”, Master's Thesis Report, ETH Zurich, 2011.

Agencja Promocji Inwestycji Spółka z o.o.: http://narew-ostroleka.eu/o_inwestycji, 2015.

Agencja Rynku Energii S.A.: “Statystyka Elektroenergetyki Polskiej 2011”, Warszawa 2012.

ELTRIM: Overhead lines wires catalogue, http://eltrim.com.pl/images/Katalog/eltrim_final.pdf, 2014.

J. D. McCalley: “The DC Power Flow Equations”, http://home.eng.iastate.edu/~jdm/ee553/DCPowerFlowEquations.pdf, Iowa State University, 2011.

K. Purchala, L. Meeus, D. Van Dommelen, and R. Belmans: “Usefulness of DC Power Flow for Active Power Flow Analysis”, IEEE Power Engineering Society General Meeting, vol. 1, pp. 454-459, 2005.

K. Van den Bergh, E. Delarue and W. D'haeseleer: “DC power flow in unit commitment Models”, TME Working Paper - Energy and Environment, 2014.

P. Kundur: Power System Stability and Control, 1994.

Polskie Sieci Elektroenergetyczne S.A.: Information regarding the power grid.

R. Rosenthal: “GAMS - A User's Guide” GAMS Development Corporation, 2013.

T. Overbye: “ECE 476 – Power System Analysis” lectures, Department of Electrical and Computer Engineering, University of Illinois.

Y. Baghzouz: “Power System Representations: Voltage-Current Relations”, http://www.egr.unlv.edu/~eebag/Power%20System%20Representations.pdf, University of Nevada, 2015.

List of figures

Figure 1. Transmission map of the Polish grid....8
Figure 2. Allocation of nodes on the map....10
Figure 3. Table linking nodes with Polish regions....11
Figure 4. Circuits displayed on the map....13
Figure 5. Scheme of transmission lines....14
Figure 6. Characteristics of transmission lines....16
Figure 7. Typical values of transmission lines parameters....17
Figure 8. Reactances of transmission lines....18
Figure 8. Power consumption in nodes....19
Figure 9. Power generation in nodes....21
Figure 10. Percentage of feasible scenarios as a function of cap and tcap....26
Figure 11. Feasible scenarios as a function of cap for tcap=1400....27
Figure 12. Feasible scenarios as a function of cap for tcap=1000....27
Figure 13. Feasible scenarios as a function of cap for tcap=500....28
Figure 14. Feasible scenarios as a function of tcap for cap=1000....28
Figure 15. Feasible scenarios as a function of tcap for cap=5000....29
Figure 16. Feasible scenarios as a function of tcap for cap=10000....29

GAMS code

The followed code was used to perform the sensitivity analysis. For the assessment of a particular scenario it is necessary to remove the sets `j` and `k`.

```
$TITLE DC grid model

SETS

n nodes /n1*n12/
l Transmission lines /l1*l22/

t scenarios /1*100/
k different capacities to install /1*2/
j different transmission capacities /1*2/

ALIAS (n,nn,m,mm);

SETS

lmap(l,n,nn) Map transmission lines to connected nodes /

l1."n1"."n2",
l2."n1"."n3",
l3."n1"."n4",
l4."n1"."n12",
l5."n2"."n4",
l6."n2"."n7",
```

```
l7."n3"."n4",
l8."n3"."n5",
l9."n3"."n12",
l10."n4"."n5",
l11."n4"."n7",
l12."n4"."n8",
l13."n4"."n9",
l14."n5"."n6",
l15."n5"."n8",
l16."n5"."n10",
l17."n5"."n11",
l18."n5"."n12",
l19."n7"."n9",
l20."n8"."n9",
l21."n9"."n10",
l22."n10"."n11"

 /

c Transmission line circuits (up to 4 circuits per line)
/c1*c4/

z Transmission line circuits (list form)
/z1*z41/

lcmap(z,l,c) Map transmission circuits to line and circuit /
z1."l1"."c1",
z2."l2"."c1",
z3."l2"."c2",
z4."l3"."c1",
```

```
z5."l4"."c1",
z6."l5"."c1",
z7."l6"."c1",
z8."l7"."c1",
z9."l7"."c2",
z10."l7"."c3",
z11."l8"."c1",
z12."l9"."c1",
z13."l10"."c1",
z14."l10"."c2",
z15."l11"."c1",
z16."l11"."c2",
z17."l11"."c3",
z18."l12"."c1",
z19."l12"."c2",
z20."l12"."c3",
z21."l13"."c1",
z22."l14"."c1",
z23."l15"."c1",
z24."l15"."c2",
z25."l15"."c3",
z26."l16"."c1",
z27."l16"."c2",
z28."l17"."c1",
z29."l17"."c2",
z30."l18"."c1",
z31."l19"."c1",
z32."l19"."c2",
z33."l20"."c1",
z34."l20"."c2",
```

```
z35."l20"."c3",
z36."l20"."c4",
z37."l21"."c1",
z38."l21"."c2",
z39."l21"."c3",
z40."l21"."c4",
z41."l22"."c1"
/

lcmap2(l,c) Map transmission circuits to line /
"l1"."c1",
"l2"."c1",
"l2"."c2",
"l3"."c1",
"l4"."c1",
"l5"."c1",
"l6"."c1",
"l7"."c1",
"l7"."c2",
"l7"."c3",
"l8"."c1",
"l9"."c1",
"l10"."c1",
"l10"."c2",
"l11"."c1",
"l11"."c2",
"l11"."c3",
"l12"."c1",
"l12"."c2",
"l12"."c3",
```

```
"l13"."c1",
"l14"."c1",
"l15"."c1",
"l15"."c2",
"l15"."c3",
"l16"."c1",
"l16"."c2",
"l17"."c1",
"l17"."c2",
"l18"."c1",
"l19"."c1",
"l19"."c2",
"l20"."c1",
"l20"."c2",
"l20"."c3",
"l20"."c4",
"l21"."c1",
"l21"."c2",
"l21"."c3",
"l21"."c4",
"l22"."c1"
/
;

PARAMETERS

volt(z)   Voltages of the circuits in kV /
z1        400
z2        400
z3        220
```

z4	220
z5	400
z6	400
z7	220
z8	220
z9	220
z10	220
z11	400
z12	220
z13	220
z14	220
z15	400
z16	220
z17	220
z18	400
z19	220
z20	220
z21	400
z22	400
z23	400
z24	400
z25	220
z26	400
z27	220
z28	400
z29	220
z30	220
z31	400
z32	220
z33	400

```
z34        400
z35        220
z36        220
z37        400
z38        400
z39        220
z40        220
z41        220
/

distline(z)   distances of the circuits in km /

z1        352.60
z2        114.00
z3        185.58
z4        376.47
z5        153.77
z6        241.26
z7        323.44
z8        100.74
z9        100.74
z10        137.86
z11        304.88
z12        270.42
z13        265.11
z14        182.93
z15        405.62
z16        320.79
z17        320.79
z18        129.91
```

```
z19        217.40
z20        190.89
z21        400.33
z22        174.98
z23        185.58
z24        220.05
z25        235.96
z26        288.98
z27        448.05
z28        106.05
z29        408.28
z30        267.77
z31        299.59
z32        474.55
z33        211.50
z34        204.14
z35        156.42
z36        182.93
z37        299.58
z38        278.37
z39        243.91
z40        182.93
z41        167.03
/

Q_node(n) Demand at node n [MW e] /
n1        1284.4
n2        908.9
n3        1340.5
n4        2075.8
```

```
n5        3320.5
n6        450.2
n7        2675.2
n8        1925.2
n9        6723.0
n10        1617.2
n11        917.7
n12        562.4
/

G_inst(n) Generation installed at node n balanced with the total demand [MW e]
;
G_inst('n1')=1238.4*(23801.0/35305.5)  ;
G_inst('n2')=2226.3*(23801.0/35305.5)  ;
G_inst('n3')=731.7*(23801.0/35305.5)  ;
G_inst('n4')=2799.9*(23801.0/35305.5)  ;
G_inst('n5')=5103.2*(23801.0/35305.5)  ;
G_inst('n6')=173.1*(23801.0/35305.5)  ;
G_inst('n7')=2915.2*(23801.0/35305.5)  ;
G_inst('n8')=5859.9*(23801.0/35305.5)  ;
G_inst('n9')=11332.2*(23801.0/35305.5)  ;
G_inst('n10')=2445.5*(23801.0/35305.5)  ;
G_inst('n11')=406.9*(23801.0/35305.5)  ;
G_inst('n12')=73.2*(23801.0/35305.5)  ;

scalar P_base base power to the per-unit system /730/;

PARAMETERS

ccap_l(z) transmission capacity of a circuit in list form
```

```
ccap_t(l,c) transmission capacity of a circuit in table form
tcap_l(l) total capacity of the line
tcap(nn,mm) total capacity of the line expressed with nodes
tcap_base(nn,mm) total capacity base to be modified;

*Transmission capacity characterization
loop(z,
         IF ((volt(z) eq 220),
              ccap_l(z)=1625.58;
             );
         IF ((volt(z) eq 400),
              ccap_l(z)=2955.6;
             );
);

loop(lcmap(z,l,c),
         ccap_t(l,c) = ccap_l(z);
);
tcap_l(l) =  sum(c$lcmap2(l,c), ccap_t(l,c));
loop(lmap(l,nn,mm),
         tcap_base(nn,mm) = tcap_l(l);
);

*------------------------------------------------------------------------------
*TRANSMISSION LINE REACTANCE CALCULATION:
*------------------------------------------------------------------------------

PARAMETERS

x(nn,mm) Line reactance from node nn to node mm
```

```
vol(l,c) voltage of circuit in table form
dist(l,c) distances in table form
xcdisohm(l,c) Circuit reactance [ohm per km]
xcohm(l,c) Circuit reactance [ohm]
xcmax Maximum circuit reactance [ohm]
xcpu(l,c) Circuit reactance [p.u.]
xpu(l) Equivalent single line reactance [p.u.]
;
loop(lcmap(z,l,c),
vol(l,c) = volt(z);
);
loop(lcmap(z,l,c),
dist(l,c) = distline(z);
);

option lmap:1:1:2;
display n, l, lmap, c, lcmap,lcmap2, vol,dist;


xcdisohm(l,c)$lcmap2(l,c) = -0.0006*vol(l,c) + 0.6029;
* Multiply by line distance :
xcohm(l,c) = xcdisohm(l,c)*dist(l,c);
* Determine the maximum circuit reactance value:
xcmax = smax((l,c),xcohm(l,c));
* Convert to per unit:
xcpu(l,c) = xcohm(l,c)/xcmax;
* Convert parallel circuit reactances to single line reactance:
xpu(l) = 1 / sum(c$lcmap2(l,c), 1/xcpu(l,c));
* Express line reactance in terms of nodes:
loop(lmap(l,nn,mm),
```

```
x(nn,mm) = xpu(l);
);
option xcdisohm:4, xcohm:2, xcmax:2, xcpu:4, xpu:4, x:4;
display xcdisohm, xcohm, xcmax, xcpu, xpu, x;

*------------------------------------------------------------------------------
*POWER TRANSMISSION MODEL:
*------------------------------------------------------------------------------


VARIABLES

Pf(nn,mm)   the power flow from nn to mm [MW e]
d(n)            the delta of node n
dummy       ;

PARAMETERS

Q(n) total demand in node n
G(n) total generation in node n;

EQUATIONS

tconspos(n,nn) Transmission capacity constraint  positive side
tconsneg(n,nn) Transmission capacity constraint  negative side

node(n)                conservation of energy in each node IN PER UNIT USING AS
BASE 730 MVAR OF THE TRANSFORMER
line(n,nn)              power flow in lines
delta0                 reference node for deltas
```

```
zeros(n,nn)           Zeros in the power flow matrix (nodes not connected)
edummy;

tconspos(n,nn).. Pf(n,nn) =l= tcap(n,nn)/P_base;
tconsneg(n,nn).. Pf(n,nn) =g= -tcap(n,nn)/P_base;

delta0 .. d('n1') =e= 0;
node(n) .. (G(n)-Q(n))/P_base =e= (sum(nn, Pf(n,nn))-sum(nn,Pf(nn,n))) ;
zeros(n,nn)$(x(n,nn)=0) .. Pf(n,nn)=e=0 ;
line(n,nn)$(x(n,nn)>0) .. Pf(n,nn)=e=(d(n)-d(nn))/x(n,nn)  ;

edummy .. dummy =e= 0;

Model grid /all/;

*Randomization of the input*

PARAMETERS

bin(n) binary random vector
ran(n) integer random vector
produ(n) product vector
New_Gen(n,t)  output with distribution of capacity in 50% of the nodes and 50%
of zeroes ON AVERAGE
Sum_produ sum of produ vector
;

Parameter Cap(k) MW to be installed;
Cap('1')=1000;
Cap('2')=2000;
```

```
Cap('3')=3000;

Cap('4')=4000;

Cap('5')=5000;

Cap('6')=6000;

Cap('7')=7000;

Cap('8')=8000;

Cap('9')=9000;

Cap('10')=10000;



Parameter var_tcap(j) variable used to modify tcap/

1        0.474

2        0.440

3        0.406

4        0.372

5        0.338

6        0.305

7        0.271

8        0.237

9        0.203

10       0.169

/;

PARAMETERS

Q_new(n,t) New load consumed in node n weighted by yearly consumption

Q_total Total load

Feasible_Location(n,t) for a certain array of scenarios returns 1 when feasible

Percentage(j,k) percentage of feasible scenarios for a certain cap and tcap;
```

```
Percentage(j,k)=0;
loop(j,
         tcap(n,nn)=tcap_base(n,nn)*var_tcap(j);
loop(k,
         Feasible_Location(n,t)=0;
         loop(t,
                 Sum_produ=0;
                 loop(n,
                         ran(n)=uniformint(1,3);
                         bin(n)=uniformint(0,1);
                         produ(n)=bin(n)*ran(n);
                         Sum_produ=Sum_produ+produ(n);

                 );
                 loop(n,
                         IF(Sum_produ ne 0,
                         New_Gen(n,t)=Cap(k)*produ(n)/Sum_produ;
                         );
                 );
         );

         Q_total=sum(n,Q_node(n));

         loop(t,
                loop(n,
                   Q_new(n,t)=Cap(k)*(Q_node(n)/Q_total);
                );

                 Q(n)=Q_node(n)+Q_new(n,t);
```

```
                G(n)=G_inst(n)+New_Gen(n,t);

                Solve grid using lp minimizing dummy;

                IF ((grid.modelstat eq 1),
                Feasible_Location(n,t)=New_Gen(n,t);
                execute_unload "results.gdx";
                Percentage(j,k)=Percentage(j,k)+1;
                );

        );

);

);

display Percentage;
execute 'gdxxrw.exe results.gdx par=Percentage rng=b3'
```